

GATE-LEVEL MODELING

- A design at a low level of abstraction gate level
- Most digital design is now done at gate level or higher levels of abstraction
- At gate level, the circuit is described in terms of gates (e.g., and, nand)
- Hardware design at this level is intuitive for a user with a basic knowledge of digital logic design because it is possible to see a one-to-one correspondence between the logic circuit diagram and the Verilog description
- Actually, the lowest level of abstraction is switch- (transistor-) level modeling
- However, with designs getting very complex, very few hardware designers work at switch level

GATE-LEVEL MODELING - LEARNING OBJECTIVES

- Identify logic gate primitives provided in Verilog
- Understand instantiation of gates, gate symbols, and truth tables for and/or and buf/not type gates
- Understand how to construct a Verilog description from the logic diagram of the circuit
- Describe rise, fall, and turn-off delays in the gate-level design
- Explain min, max, and typ delays in the gate-level design


GATE TYPES

- A logic circuit can be designed by use of logic gates
- Verilog supports basic logic gates as predefined primitives
- These primitives are instantiated like modules except that they are predefined in Verilog and do not need a module definition
- All logic circuits can be designed by using basic gates
- There are two classes of basic gates:
 - and/or gates and
 - buf/not gates

And/Or Gates

- And/or gates have one scalar output and multiple scalar inputs
- The first terminal in the list of gate terminals is an output and the other terminals are inputs
- The output of a gate is evaluated as soon as one of the inputs changes
- The and/or gates available in Verilog are shown below
 - and or xor
 - nand nor xnor

Figure 1. Basic Gates



- We consider gates with two inputs. The output terminal is denoted by out. Input terminals are denoted by i1 and i2

Example 1: Gate Instantiation of And/Or Gates

1. wire OUT, IN1, IN2;
2. // basic gate instantiations.
3. and a1(OUT, IN1, IN2);
4. nand na1(OUT, IN1, IN2);
5. or or1(OUT, IN1, IN2);
6. nor nor1(OUT, IN1, IN2);
7. xor x1(OUT, IN1, IN2);
8. xnor nx1(OUT, IN1, IN2);
9. // More than two inputs; 3 input nand gate
10. nand na1_3inp(OUT, IN1, IN2, IN3);
11. // gate instantiation without instance name
12. and (OUT, IN1, IN2); // legal gate instantiation

TRUTH TABLES FOR THESE GATES

- The truth tables for these gates define how outputs for the gates are computed from the inputs
- Truth tables are defined assuming two inputs
- Outputs of gates with more than two inputs are computed by applying the truth table iteratively

Table 1. Truth Tables for And/Or

	and	i1			
		0	1	x	z
i2	0	0	0	0	0
	1	0	1	x	x
	x	0	x	x	x
	z	0	x	x	x

	nand	i1			
		0	1	x	z
i2	0	1	1	1	1
	1	1	0	x	x
	x	1	x	x	x
	z	1	x	x	x

	or	i1			
		0	1	x	z
i2	0	0	1	x	x
	1	1	1	1	1
	x	x	1	x	x
	z	x	1	x	x

	nor	i1			
		0	1	x	z
i2	0	1	0	x	x
	1	0	0	0	0
	x	x	0	x	x
	z	x	0	x	x

	xor	i1			
		0	1	x	z
i2	0	0	1	x	x
	1	1	0	x	x
	x	x	x	x	x
	z	x	x	x	x

	xnor	i1			
		0	1	x	z
i2	0	1	0	x	x
	1	0	1	x	x
	x	x	x	x	x
	z	x	x	x	x

Gates

Buf/Not Gates

- Buf/not gates have one scalar input and one or more scalar outputs
- The last terminal in the port list is connected to the input. Other terminals are connected to the outputs
- Two basic buf/not gate primitives are provided in Verilog
 - buf not



Example 2: Gate Instantiations of Buf/Not Gates

1. `// basic gate instantiations.`
2. `buf b1(OUT1, IN);`
3. `not n1(OUT1, IN);`

4. `// More than two outputs`
5. `buf b1_2out(OUT1, OUT2, IN);`

6. `// gate instantiation without instance name`
7. `not (OUT1, IN); // legal gate instantiation`

Truth Tables for Buf/Not Gates

Table 2. Truth Tables for Buf/Not Gates

buf	in	out
	0	0
	1	1
	x	x
	z	x


not	in	out
	0	1
	1	0
	x	x
	z	x

Gates with an additional control signal on buf and not gates

■ Bufif/notif

- bufif1 notif1
 - bufif0 notif0
- These gates are used when a signal is to be driven only when the control signal is asserted
 - Such a situation is applicable when multiple drivers drive the signal
- These gates propagate only if their control signal is asserted
 - They propagate z if their control signal is deasserted
 - Symbols for bufif/notif are shown in Figure 3

Figure 3. Gates Bufif and Notif



Truth Tables for Bufif/Notif Gates

Table 3. Truth Tables for Bufif/Notif Gates

		ctrl			
bufif1		0	1	x	z
in	0	z	0	L	L
	1	z	1	H	H
	x	z	x	x	x
	z	z	x	x	x

		ctrl			
bufif0		0	1	x	z
in	0	0	z	L	L
	1	1	z	H	H
	x	x	z	x	x
	z	x	z	x	x

		ctrl			
notif1		0	1	x	z
in	0	z	1	H	H
	1	z	0	L	L
	x	z	x	x	x
	z	z	x	x	x

		ctrl			
notif0		0	1	x	z
in	0	1	z	H	H
	1	0	z	L	L
	x	x	z	x	x
	z	x	z	x	x

Example 3: Gate Instantiations of Bufif/Notif Gates

- These drivers are designed to drive the signal on mutually exclusive control signals.
- Example 3 shows examples of instantiation of bufif and notif gates.

- **Example 3 Gate Instantiations of Bufif/Notif Gates**

1. //Instantiation of bufif gates.
2. bufif1 b1 (out, in, ctrl);
3. bufif0 b0 (out, in, ctrl);
4. //Instantiation of notif gates
5. notif1 n1 (out, in, ctrl);
6. notif0 n0 (out, in, ctrl);

Array of Instances

- There are many situations when repetitive instances are required
- These instances differ from each other only by the index of the vector to which they are connected
- To simplify specification of such instances, Verilog HDL allows an array of primitive instances to be defined
- Example 4 shows an example of an array of instances

Example 4: Simple Array of Primitive Instances

1. **Example 4 Simple Array of Primitive Instances**
2. `wire [7:0] OUT, IN1, IN2;`
3. `// basic gate instantiations.`
4. `nand n_gate[7:0](OUT, IN1, IN2);`
5. `// This is equivalent to the following 8 instantiations`
6. `nand n_gate0(OUT[0], IN1[0], IN2[0]);`
7. `nand n_gate1(OUT[1], IN1[1], IN2[1]);`
8. `nand n_gate2(OUT[2], IN1[2], IN2[2]);`
9. `nand n_gate3(OUT[3], IN1[3], IN2[3]);`
10. `nand n_gate4(OUT[4], IN1[4], IN2[4]);`
11. `nand n_gate5(OUT[5], IN1[5], IN2[5]);`
12. `nand n_gate6(OUT[6], IN1[6], IN2[6]);`
13. `nand n_gate7(OUT[7], IN1[7], IN2[7]);`

Gate-level multiplexer

- We will design a 4-to-1 multiplexer with 2 select signals
- Multiplexers serve a useful purpose in logic design
- They can connect two or more sources to a single destination
- They can also be used to implement boolean functions
- We will assume for this example that signals s1 and s0 do not get the value x or z
- The I/O diagram and the truth table for the multiplexer are shown in Figure 4
- The I/O diagram will be useful in setting up the port list for the multiplexer

Figure 4. 4-to-1 Multiplexer





Figure 5. Logic Diagram for Multiplexer



- The logic diagram has a one-to-one correspondence with the Verilog description
- The Verilog description for the multiplexer is shown in Example 5
- Two intermediate nets, `s0n` and `s1n`, are created; they are complements of input signals `s1` and `s0`
- Internal nets `y0`, `y1`, `y2`, `y3` are also required
- Note that instance names are not specified for primitive gates, `not`, `and`, and `or`
- Instance names are optional for Verilog primitives but are mandatory for instances of user-defined modules

Example 5.1: Verilog Description of Multiplexer

```
1. // Module 4-to-1 multiplexer. Port list is taken exactly from the I/O diagram.
2. module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);
3. // Port declarations from the I/O diagram
4. output out;
5. input i0, i1, i2, i3;
6. input s1, s0;
7. // Internal wire declarations
8. wire s1n, s0n;
9. wire y0, y1, y2, y3;
12.// Gate instantiations
13.// Create s1n and s0n signals.
14.not (s1n, s1);
15.not (s0n, s0);
16.// 3-input and gates instantiated
17.and (y0, i0, s1n, s0n);
18.and (y1, i1, s1n, s0);
19.and (y2, i2, s1, s0n);
20.and (y3, i3, s1, s0);
21.// 4-input or gate instantiated
22.or (out, y0, y1, y2, y3);
23.endmodule
```

Example 5.2 Stimulus for Multiplexer

1. `// Define the stimulus module (no ports)`
2. `module stimulus;`
3. `// Declare variables to be connected`
4. `// to inputs`
5. `reg IN0, IN1, IN2, IN3;`
6. `reg S1, S0;`
7. `// Declare output wire`
8. `wire OUTPUT;`
9. `// Instantiate the multiplexer`
10. `mux4_to_1 mymux(OUTPUT, IN0, IN1, IN2, IN3, S1, S0);`

```
11. // Stimulate the inputs

12. // Define the stimulus module (no ports)

13. initial

14. begin

15. // set input lines

16. IN0 = 1; IN1 = 0; IN2 = 1; IN3 = 0;

17. #1 $display("IN0= %b, IN1= %b, IN2= %b, IN3= %b\n",IN0,IN1,IN2,IN3);

18. // choose IN0

19. S1 = 0; S0 = 0;

20. #1 $display("S1 = %b, S0 = %b, OUTPUT = %b\n", S1, S0, OUTPUT);

21. // choose IN1

22. S1 = 0; S0 = 1;

23. #1 $display("S1 = %b, S0 = %b, OUTPUT = %b\n", S1, S0, OUTPUT);
```

```
24. // choose IN2

25. S1 = 1; S0 = 0;

26. #1 $display("S1 = %b, S0 = %b, OUTPUT = %b \n", S1, S0, OUTPUT);

27. // choose IN3

28. S1 = 1; S0 = 1;

29. #1 $display("S1 = %b, S0 = %b, OUTPUT = %b \n", S1, S0, OUTPUT);

30. end

31. endmodule
```

- The output of the simulation is shown below
- Each combination of the select signals is tested
 1. $IN0 = 1, IN1 = 0, IN2 = 1, IN3 = 0$
 2. $S1 = 0, S0 = 0, OUTPUT = 1$
 3. $S1 = 0, S0 = 1, OUTPUT = 0$
 4. $S1 = 1, S0 = 0, OUTPUT = 1$
 5. $S1 = 1, S0 = 1, OUTPUT = 0$

4-BIT RIPPLE CARRY FULL ADDER


- In this example, we design a 4-bit full adder
- We use primitive logic gates, and we apply stimulus to the 4-bit full adder to check functionality
- For the sake of simplicity, we will implement a ripple carry adder
- The basic building block is a 1-bit full adder
- The mathematical equations for a 1-bit full adder are shown below

$$\text{sum} = (a \oplus b \oplus \text{cin})$$

$$\text{cout} = (a * b) + \text{cin} * (a \oplus b)$$

Example 7: Verilog Description for 1-bit Full Adder


Figure 6. 1-bit Full Adder



```
1. // Define a 1-bit full adder
2. module fulladd(sum, c_out, a, b, c_in);
3. // I/O port declarations
4. output sum, c_out;
5. input a, b, c_in;
6. // Internal nets
7. wire s1, c1, c2;
8. // Instantiate logic gate primitives
9. xor (s1, a, b);
10. and (c1, a, b);
11. xor (sum, s1, c_in);
12. and (c2, s1, c_in);
13. or (c_out, c2, c1);
14. endmodule
```

A 4-BIT RIPPLE CARRY FULL ADDER CONSTRUCTED FROM FOUR 1-BIT FULL ADDERS

Figure 7. 4-bit Ripple Carry Full Adder



- This structure can be translated to Verilog as shown in Example 8.1
- Note that the port names used in a 1-bit full adder and a 4-bit full adder are the same but they represent different elements.
- The element sum in a 1-bit adder is a scalar quantity and the element sum in the 4-bit full adder is a 4-bit vector quantity
- Verilog keeps names local to a module. Names are not visible outside the module unless hierarchical name referencing is used
- Also note that instance names must be specified when defined modules are instantiated, but when instantiating Verilog primitives, the instance names are optional

EXAMPLE 8.1: VERILOG DESCRIPTION FOR 4-BIT RIPPLE CARRY FULL ADDER

```
1. // Define a 4-bit full adder
2. module fulladd4(sum, c_out, a, b, c_in);
3. // I/O port declarations
4. output [3:0] sum;
5. output c_out;
6. input[3:0] a, b;
7. input c_in;
8. // Internal nets
9. wire c1, c2, c3;
10. // Instantiate four 1-bit full adders.
11. fulladd fa0(sum[0], c1, a[0], b[0], c_in);
12. fulladd fa1(sum[1], c2, a[1], b[1], c1);
13. fulladd fa2(sum[2], c3, a[2], b[2], c2);
14. fulladd fa3(sum[3], c_out, a[3], b[3], c3);
15. endmodule
```

Example 8.2 Stimulus for 4-bit Ripple Carry Full Adder

```
1. // Define the stimulus (top level module)
2. module stimulus;
3. // Set up variables
4. reg [3:0] A, B;
5. reg C_IN;
6. wire [3:0] SUM;
7. wire C_OUT;
8. // Instantiate the 4-bit full adder. call it FA1_4
9. fulladd4 FA1_4(SUM, C_OUT, A, B, C_IN);
10. // Set up the monitoring for the signal values
11. initial
12. begin
13. $monitor($time," A= %b, B=%b, C_IN= %b, --- C_OUT= %b, SUM= %b\n", A, B, C_IN, C_OUT, SUM);
14. end
15. // Stimulate inputs
16. initial
17. begin
18. A = 4'd0; B = 4'd0; C_IN = 1'b0;
19. #5 A = 4'd3; B = 4'd4;
20. #5 A = 4'd2; B = 4'd5;
21. #5 A = 4'd9; B = 4'd9;
22. #5 A = 4'd10; B = 4'd15;
23. #5 A = 4'd10; B = 4'd5; C_IN = 1'b1;
24. end
25. endmodule
```

- The output of the simulation is shown below
- 1. 0 A= 0000, B=0000, C_IN= 0, --- C_OUT= 0, SUM= 0000
- 2. 5 A= 0011, B=0100, C_IN= 0, --- C_OUT= 0, SUM= 0111
- 3. 10 A= 0010, B=0101, C_IN= 0, --- C_OUT= 0, SUM= 0111
- 4. 15 A= 1001, B=1001, C_IN= 0, --- C_OUT= 1, SUM= 0010
- 5. 20 A= 1010, B=1111, C_IN= 0, --- C_OUT= 1, SUM= 1001
- 6. 25 A= 1010, B=0101, C_IN= 1,, C_OUT= 1, SUM= 0000